<!--#Include File="conn.asp" -->
<!--#Include File="include/inc.asp"-->
<%
'====================================================================
'= Team Elite - Elite Article System
'= Copyright (c) 2005 - 2006 Eason Chan All Rights Reserved.
'=-------------------------------------------------------------------
'= 版权协议：
'=	GPL (The GNU GENERAL PUBLIC LICENSE Version 2, June 1991)
'=-------------------------------------------------------------------
'= 文件名称：Member_List.asp
'= 摘    要：会员列表文件
'=-------------------------------------------------------------------
'= 最后更新：eason007
'= 最后日期：2005-06-05
'====================================================================

Dim Action
Dim PageNum,PageSize,PageCount,ReCount
Dim FieldName(1),FieldValue(1)
Dim QueryArray
Dim PageContent,QueryList
Dim i

Action=Request.QueryString ("action")
PageNum=EA_Pub.SafeRequest(1,"page",0,1,0)
PageSize=20

FieldName(0)="action"
FieldValue(0)=Action

ReCount=EA_DBO.Get_Member_Total()(0,0)
PageCount=EA_Pub.Stat_Page_Total(PageSize,ReCount)
If PageNum>PageCount And PageCount>0 Then PageNum=PageCount

QueryArray=EA_DBO.Get_Member_List(Action,"",PageNum,PageSize)
If IsArray(QueryArray) Then 
	'QueryList="<table border=""0"" align=""center"" cellpadding=""0"" cellspacing=""0"" width=""98%"" style=""word-break: break-all"">"
	For i=0 To UBound(QueryArray,2)
		QueryList=QueryList&"<tr align=""center"" height=""22"">"
		'name
		QueryList=QueryList&"<td>"&QueryArray(1,i)&"</td>"
		'sex
		QueryList=QueryList&"<td>"&QueryArray(2,i)&"</td>"
		'email
		QueryList=QueryList&"<td><a href=""mailto:"&QueryArray(3,i)&""">"&QueryArray(3,i)&"</a></td>"
		'qq number
		QueryList=QueryList&"<td>"&QueryArray(4,i)&"</td>"
		'group name
		QueryList=QueryList&"<td>"&QueryArray(5,i)&"</td>"
		'regtime
		QueryList=QueryList&"<td>"&FormatDateTime(QueryArray(6,i),2)&"</td>"
		'home page
		QueryList=QueryList&"<td>"&QueryArray(7,i)&"</td>"
		'post total
		QueryList=QueryList&"<td>"&QueryArray(8,i)&"</td>"
		QueryList=QueryList&"</tr>"
	Next
	QueryList=QueryList&"<tr>"
	QueryList=QueryList&"<td align=""right"" colspan=""8"" bgcolor=""efefef"">"&EA_Temp.PageList(PageCount,PageNum,FieldName,FieldValue)&"</td>"
	QueryList=QueryList&"</tr>"
	'QueryList=QueryList&"</table>"
End If

PageContent=EA_Temp.Load_Template(0,"memberlist")

EA_Temp.Title=EA_Pub.SysInfo(0)&" - 会员列表"
EA_Temp.Nav="<a href=""./""><b>"&EA_Pub.SysInfo(0)&"</b></a> - 会员列表"

PageContent=Replace(PageContent,"{$Query_List$}",QueryList)

PageContent=EA_Temp.Replace_PublicTag(PageContent)

Response.Write PageContent

Call EA_Pub.Close_Obj
Set EA_Pub=Nothing
%>